
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of IMS test case 19.5.7 which is part of the IMS test suite in the ‘iwd-TTCN-B2013-03_D14wk10’ delivery.

The test case can be demonstrated to run with one LTE UE supporting IMS Emergency Call (see section 5) on different Test Platforms. Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_19_5_7
Test Group:
IMS_36523\...\IMS\19\IMS_CC_EmergencyCallTestcases.ttcn
ATS Version:
iwd-TTCN-B2013-03_D14wk10
System Simulator used:
Anritsu Protocol Conformance Test System ME7832L
UE used:
Intel XMM7160
Verification Status:
PASS


4. Corrections required for IMS test case 19.5.7
This section describes the TTCN changes required to ‘iwd-TTCN-B2013-03_D14wk10’ release
Change 1 – Corrections to function ”f_TC_19_5_7_IMS2()”
	Function name
	f_TC_19_5_7_IMS2()  

	Reason for change
	1. According to spec. 34.229  section 19.5.7.4 in Step 2: “However during the steps specified in Annex C.22 the SS shall delay every response sent to the UE by one second, to cause the half of the emergency registration time to expire after step 3 of Annex C.22 (180 response sent by the SS).” But In this case UE sends re-Register request ater step 1 of Annex C.22. After correction of delay to 0.5s re-Register request is sent in expected time (after step 3 of Annex C.22)
A draft prose CR will be raised at next RAN5#63 meeting

2. Registration expire timer t_RegExpireMax (5 seconds) need to be started after Register request to IMS Emergency services is received and be stopped after re-Register request is successfully received by SS.
3. According to spec 34.229 section 19.5.7.4 parallel behaviour consist of re-registration procedure instead of Initial Emergency Registration.
4. As Emergency Call initiation was not completed (no active emergency call), emergency call release is not needed 

	Summary of change
	1. Change v_DelayBeforeSending from 1 second to 0.5 second

2. Use content of f_IMS_EmergencyRegistration_C20 and start timer t_RegExpireMax (5 seconds) after execution of function f_IMS_REGISTER_InitialRequest to ensure that timer for re-registration is started also on SS side
3. For parallel behaviour use procedure for IMS emergency re-registration

4. Delete f_IMS_EmergencyCallRelease and revert change back to v_InviteRequest and delete line v_InviteRequest := v_Invite_DATA_REQ.Request.Invite

	TTCN module
	IMS/19/IMS_CC_EmergencyCallTestcases.ttcn

	MCC160 Comment
	


Before change

function f_TC_19_5_7_IMS2() runs on IMS_PTC

  { /* IMS implementation for PDN2 - Emergency call */

    var INVITE_Request v_InviteRequest;

    var REGISTER_Request v_RegisterRequest;

    var integer v_Expiry := 5;     // 5 sec used per step 1 of the Expected sequence

    var float v_ReRegistrationTime := int2float(v_Expiry) / 2.0;
    var float v_DelayBeforeSending := 1.0; // as per step 2 of the Expected sequence

    timer t_RegExpireMax := f_IMS_SetTimerToleranceMax(v_ReRegistrationTime); // more than half of the expiration time

    f_IMS_PTC_Init(PDN_2);

    f_IMS_IMS_WaitForTrigger(OtherPDN);

    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NormalService); // steps 1-15 of 4.5A.4

    f_IMS_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    v_RegisterRequest := f_IMS_EmergencyRegistration_C20(-, int2str(v_Expiry)); // Initial condition: Emergency registration, expiration time is set to 5 seconds

    t_RegExpireMax.start; // more than half of the expiration time shall elapse

    //@siclog "Step 2" siclog@

    v_InviteRequest := f_IMS_EmergencyCallSetup_AnnexC22_Steps2to3_Delayed(-, v_DelayBeforeSending); // prose unclear, inclusion of step 4 to be discussed, would need further changes Q3 to OB

    //@siclog "Step 1-4 Parallel behaviour" siclog@

    // Parallel behaviour Steps 1 to 4 re-registration

    f_IMS_EmergencyRegistration_C20();
    t_RegExpireMax.stop;
    f_IMS_TestBody_Set(false);

    f_IMS_EmergencyCallRelease(v_InviteRequest); // Q4 to OB Invite still valid ??

    f_IMS_IMS_SendCoOrdMsg(OtherPDN); // IMS call has been released

  }
After change

function f_TC_19_5_7_IMS2() runs on IMS_PTC

  { /* IMS implementation for PDN2 - Emergency call */

    var INVITE_Request v_InviteRequest;

    var REGISTER_Request v_RegisterRequest;

    var integer v_Expiry := 5;     // 5 sec used per step 1 of the Expected sequence

    var float v_ReRegistrationTime := int2float(v_Expiry);                        

    var float v_DelayBeforeSending := 0.5; // as per step 2 of the Expected sequence

    timer t_RegExpireMax := f_IMS_SetTimerToleranceMax(v_ReRegistrationTime); // more than half of the expiration time

    var IMS_DATA_REQ v_IMS_DATA_REQ; 

    var IMS_DATA_REQ v_Invite_DATA_REQ;

    var integer v_NonceCount;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var template (present) SipUrl v_SipUrl; 

    var template (value) IMS_SecurityInfo_Type v_SecurityInfo;

    var SecurityClientParams_Type v_SecurityClientParams;
    f_IMS_PTC_Init(PDN_2);

    f_IMS_IMS_WaitForTrigger(OtherPDN);

    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NormalService); // steps 1-15 of 4.5A.4

    f_IMS_TestBody_Set(true);    

    //@siclog "Step 1" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, -, false, true);         

    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;

    t_RegExpireMax.start;

    f_IMS_Register_SecurityInit(v_RegisterRequest);

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterRequest))));

    v_InviteRequest := f_IMS_REGISTER_SubsequentRequest(-, true);

    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterRequest, true, int2str(v_Expiry))))); // @sic R5s130561 sic@
    //@siclog "Step 2" siclog@

    v_Invite_DATA_REQ := f_IMS_EmergencyCallSetup_AnnexC22_Steps2to3_Delayed(-, v_DelayBeforeSending); // prose unclear, inclusion of step 4 to be discussed, would need further changes Q3 to OB

    v_InviteRequest := v_Invite_DATA_REQ.Request.Invite;

    v_SipUrl := cr_SipUri_HostPort(f_IMS_PTC_ImsInfo_GetHomeDomainName());
    //@siclog "Step 1-4 Parallel behaviour" siclog@

    // Parallel behaviour Steps 1 to 4 re-registration

    alt {  

  [] IMS_Server.receive(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl))) -> value v_IMS_DATA_REQ

   {  

    t_RegExpireMax.stop;

    v_NonceCount := 2;

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(v_IMS_DATA_REQ, true, v_NonceCount);

    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;

    v_SecurityClientParams := f_IMS_PTC_ReRegistration_SecurityClientParamsGet();

    f_IMS_PTC_ReRegistration_SecurityClientParamsCheck(v_RegisterRequest, v_SecurityClientParams); 

    // store old security info to release it after re-authentication @sic R5130900 - MCC160 implementation sic@

    v_SecurityInfo := cs_IMS_SecurityInfo(f_IMS_PTC_Security_GetProtectedPorts(),

                                          f_IMS_PTC_Security_GetSPIs(),

                                          f_IMS_PTC_Security_GetIntegrityAlgorithm(),

                                          f_IMS_PTC_Security_GetCipheringAlgorithm());

    f_IMS_Register_SecurityInit(v_RegisterRequest, not4b tsc_IMS_AuthRAND);    /* generate a new RAND */

    v_RoutingInfo := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo,

                                     cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterRequest))));

    v_NonceCount := 1; // nonce counter is reset to 1 due to re-authentication

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, true, v_NonceCount);

    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;

    f_IMS_Register_SecurityRelease(v_SecurityInfo);

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo),

                                     cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterRequest, true, "1200"))));

   }

  }  

    f_IMS_TestBody_Set(false);

    f_IMS_IMS_SendCoOrdMsg(OtherPDN); // IMS call has been released

  }

5. Execution Log Files
INTEL UE

The Intel XMM7160 UE passed this test case on Anritsu ME7832L LTE System in LTE band 1. The documentation below is enclosed as evidence of the successful test case run [1]:
Log masking as per RAN5 PRD12 was applied to prevent any disclosure of confidential UE information.

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

	R5s140339:    Supporting information for agreement of IMS TC 19.5.7 in LTE FDD mode.
                          This archive comprises:

                        -  html and text format execution log files


